
Hinweis: Bei den folgenden Aufgaben bietet es sich an, den Ausgabestopp zu aktivieren, so 
dass der MOPS jedes Mal anhält, wenn etwas ins Ausgaberegister geschoben wird. Weiter 
geht es dann immer mit der Leertaste.

Später kann die Animation auch auf aus gestellt werden.

!

M O P S

MOdellrechner mit PSeudoAssembler

Aufgabensammlung mit Lösungen

MOPS – Übersicht Befehle und Syntax 

!  

!  

!  



Einfache Addition

Schreibe ein Programm, das die Zahlen 19 und 12 addiert und das Ergebnis ins 
Ausgaberegister schiebt.

Hinweis, wenn Du nicht weiterkommst:
• Lade zunächst die 19 in den Akku(mulator)
• Speichere den Akkuinhalt in der Zelle mit der Adresse $64
• Lade nun die 12 in den Akku
• Addiere den Inhalt aus Zelle $64 dazu
• Speichere den (neuen) Akkuinhalt in Zelle $65
• Gib den Inhalt von $65 aus

Lösung Einfache Addition

ld 19
st $64
ld 12
add $64
st $65
out $65
end

Kommentar: Zum Nachvollziehen, wie ein Programm übersetzt wird und wie der MOPS arbeitet.

EINE ERFOLGTE EINFÜHRUNG IN DIE VON-NEUMANN-ARCHITEKTUR UND IN DAS ARBEITEN MIT DEM MOPS 
WIRD HIER VORAUSGESETZT!  



Addition zweier Zahlen

Schreibe ein Programm, das zwei Zahlen einliest, diese addiert und das Ergebnis ins 
Ausgaberegister schiebt.

Beispiel: A = 4, B = 6 Ausgabe: 10,   da 4 + 6 = 10

Lösung Addition zweier Zahlen

in $64
in $65
ld $64
add $65
st $66
out $66
end



Aufgabe Naive Division

Schreibe ein Programm, das zwei Eingaben erwartet (Dividend und Divisor) und das Ergebnis 
der ganzzahligen Division im Ausgaberegister ausgibt.

Beispiel: A = 9, B = 4 Ausgabe: 2,   da 9 / 4 = 2 (ganzzahlig, Rest 1)

Läuft das Programm immer fehlerfrei? Versuche eine Situation zu erzeugen, in der es zu 
einem schweren Fehler kommt!

Lösung Naive Division

in $64
in $65
ld $64
div $65
st $66
out $66
end

Kommentar: Bei einer Division durch 0 kommt es zu einem Fehler. Über die Prüfung, ob der 
Divisor 0 ist, kann das Konzept der Entscheidungsanweisung eingeführt werden.



Aufgabe Division mit Prüfung

Schreibe ein Programm, das zwei Eingaben erwartet (Dividend und Divisor) und das Ergebnis 
der ganzzahligen Division im Ausgaberegister ausgibt. Dabei sollen ungültige Eingaben 
nicht mehr möglich sein, es muss also auf Division durch 0 geprüft werden.

Beispiel: A = 9, B = 4 Ausgabe: 2,   da 9 / 4 = 2 (ganzzahlig, Rest 1)

A = 8, B = 0 Programm wird beendet

EXKURS zum Modulo:

Der Modulo ist der ganzzahlige Rest einer ganzzahligen Division, hier also immer der 
Rest:

8 / 5 = 1, Rest 3 
8 / 4 = 2, Rest 0
8 / 3 = 2, Rest 2

also:

8 mod 5 = 3
8 mod 4 = 0
8 mod 3 = 2

usw.

Lösung Division mit Prüfung

in $64
in $65
ld $65
cmp 0
jeq M1
ld $64
div $65
st $66
out $66
end :M1

Kommentar: Gegenüberstellung Quelltext und Struktogramm:

in $64
in $65
ld $65
cmp 0
jeq M1
ld $64
div $65
st $66
out $66
end :M1 !



Aufgabe Division mit Rest

Schreibe ein Programm, das zwei Eingaben erwartet (Dividend und Divisor) und zuerst das 
Ergebnis der ganzzahligen Division und danach den Rest der ganzzahligen Division im 
Ausgaberegister ausgibt. Zudem sollen diese beiden Zahlen (Ergebnis und Rest) in den 
Speicherzellen $66 und $67 gespeichert werden. 

Beispiel: A = 9, B = 4 Ausgabe: 2 ... 1,   da 9 / 4 = 2, Rest 1

A = 8, B = 5 Ausgabe: 1 ... 3,   da 8 / 5 = 1, Rest 3

Lösung Division mit Rest

in $64
in $65
;Ergebnis
ld $64
div $65
st $66
;Rest
ld $64
mod $65
st $67
out $66
out $67
end

Kommentar: Einführungsmöglichkeit bzw. Wiederauffrischung für den Umgang mit dem Modulo. 
Auch mit Prüfung auf Division durch 0 kombinierbar.



Aufgabe Gerade oder Ungerade

Schreibe ein Programm, das feststellt, ob eine Zahl gerade oder ungerade ist. Das 
Programm erwartet eine Eingabe X und gibt im Ausgaberegister 0 aus, wenn die Zahl gerade 
ist und 1, wenn sie ungerade ist.

Beispiel: Eingabe: 3 Ausgabe: 1

Eingabe: 8 Ausgabe: 0

Lösung Gerade oder Ungerade

in $64
ld $64
mod 2
st $65
out $65
end

Kommentar: Weitere Einführungsmöglichkeit bzw. Wiederauffrischung für den Umgang mit dem 
Modulo. Programm gibt 1 für ungerade und 0 für gerade aus. 



Aufgabe Signum-Funktion

Ein Programm soll entscheiden, ob eine Zahl negativ, 0 oder positiv ist. Ist die Zahl 
negativ, soll -1 ausgegeben werden. Bei 0 entsprechend 0 und wenn die Zahl positiv ist, 
soll 1 ausgegeben werden.

Beispiel: Eingabe: -65 Ausgabe: -1
Eingabe: 0 Ausgabe: 0
Eingabe: 45 Ausgabe: 1

Lösung Signum-Funktion

Kommentar: Hier sind auch Alternativen denkbar mit verschachtelten Entscheidungs-
anweisungen. In der hier verwendeten Version kann auch über eine Fallunterscheidung mit 
„sonst“-Teil diskutiert werden und über die Konsequenzen, wenn Fälle vergessen werden. 
Auch kann auf die switch-Anweisung vorgegriffen werden.

in $64
ld $64
cmp 0
jeq M1
jlt M2
jgt M3
out 0 :M1
jmp MEnd
out -1 :M2
jmp MEnd
out 1 :M3
end :MEnd

in $64
ld $64
cmp 0
jeq M1
jlt M2
jgt M3
out 0 :M1
jmp MEnd
out -1 :M2
jmp MEnd
out 1 :M3
end :MEnd

!



Aufgabe Dreigeteilte Funktion

Schreibe ein Programm, das die folgende Funktion f(x) umsetzt und für einen eingegebenen 
Wert x den entsprechenden y-Wert ausgibt:

 x, für x < 0

f(x) =  0, für (x >= 0) und (x <= 5)

x2, für x > 5

Der Funktionsgraph im Bereich [-10..10] sieht dementsprechend wie folgt aus:

!

Lösung Dreigeteilte Funktion

in $64
ld $64
cmp 5
jgt M1
cmp 0
jlt M2

; sonst
ld 0
st $64
jmp M2

; > 5
mul $64 :M1
st $64
jmp M2

; Ausgabe
out $64 :M2
end

in $64
ld $64
cmp 5
jgt M1
cmp 0
jlt M2

; sonst
ld 0
st $64
jmp M2

; > 5
mul $64 :M1
st $64
jmp M2

; Ausgabe
out $64 :M2
end

!



Aufgabe Zähler

Schreibe ein Programm, das von 1 bis 9999 zählt (mehr geht beim MOPS nicht). Die Ausgabe 
soll im Ausgaberegister erfolgen.

Beispiel: 1 ... 2 ... 3 ... 4 ... usw.

Hier bietet sich dann diese Einstellung an: 

!

Lösung Zähler

Kommentar: simple Einführung der Wiederholungsanweisung 

ld 1
add 1 :M1
st $64
out $64
jmp M1
end

ld 1
add 1 :M1
st $64
out $64
jmp M1
end

!



Aufgabe Reihe aufsagen

Schreibe ein Programm, das eine ganze Zahl erwartet. Es soll dann die Zahlenreihe dieser 
Zahl automatisch immer weiter fortsetzen.

Beispiel: Eingabe: 3 Ausgabe: 3, 6, 9, 12, 15, 18 , ...

Eingabe: 4 Ausgabe: 4, 8, 12, 16, 20, ...

Lösung Reihe aufsagen

in $64
out $64
ld $64
add $64 :M1
st $65
out $65
jmp M1
end

; Alternative
ld 0
st $65
in $64
ld $65 :M1
add $64
st $65
out $65
jmp M1
end

ld 0
st $65
in $64
ld $65 :M1
add $64
st $65
out $65
jmp M1
end !



Aufgabe Summe aller Zahlen

Schreibe ein Programm, das die Summe aller Zahlen zwischen einer Startzahl und einer 
Endzahl berechnet. Dazu sollen zunächst zwei Zahlen ins Eingaberegister gelesen werden: A 
und B. Danach soll die Summe aller Zahlen von A bis B im Ausgaberegister ausgegeben 
werden.

Beispiel: A = 3, B = 8 Ausgabe: 33,   da 3 + 4 + 5 + 6 + 7 + 8 = 33

A = 12, B = 13 Ausgabe: 25,   da 12 + 13 = 25

Lösung Summe aller Zahlen

Kommentar: Alternativ kann hier die Wiederholungsanweisung zunächst auch als 
Endlosschleife mit einem Break-Befehl eingeführt werden. Dann sollte jedoch der Begriff 
„bedingte Schleife“ eingeführt werden. 

ld 0
st $66
in $64 ;A
in $65 ;B
ld $65 :M2
cmp $64
jlt M1
ld $64
add $66
st $66
ld $64
add 1
st $64
jmp M2
out $66 :M1
end

ld 0
st $66
in $64 ;A
in $65 ;B
ld $65 :M2
cmp $64
jlt M1
ld $64
add $66
st $66
ld $64
add 1
st $64
jmp M2
out $66 :M1
end

!



Aufgabe Dezimalzahl in Dualzahl

Schreibe ein Programm, das eine Dezimalzahl in eine 4-Bit Dualzahl umwandelt und 
ziffernweise über das Ausgaberegister ausgibt. Dabei wird das Problem soweit vereinfacht, 
dass die Dualzahl von rechts nach links ausgegeben werden darf.

Beispiel: Eingabe: 13 (1101) Ausgabe: 1 ... 0 ... 1 ... 1

Eingabe: 8  (1000) Ausgabe: 0 ... 0 ... 0 ... 1

Das geht mit dem nachfolgenden Algorithmus relativ simpel:

!

Hinweis: Es reicht hier, dies für die Zahlen von 0 bis 15 zu machen.

Lösung Dezimalzahl in Dualzahl     ; Alternative

Kommentar: Gegenüberstellung Quelltext und Struktogramm:

in $64
ld $64 :M1
mod 2
st $65
out $65
ld $64
div 2
st $64
cmp 0
jeq M2
jmp M1
end :M2

in $64

; KOPF
ld $64 :M1
mod 2
st $65
out $65
ld $64
div 2
st $64
cmp 0
jeq M2
jmp M1
out 0 :M2
; ENDE

end

in $64
ld $64 :M1
mod 2
st $65
out $65
ld $64
div 2
st $64
cmp 0
jeq M2
jmp M1
end :M2

!



Aufgabe Multiplikation durch Addition

Alle Grundrechenarten können auf die Addition zurückgeführt werden. So kann man die 
Subtraktion, die Multiplikation und die Division auch durch (mehrere) Additionen 
ausdrücken. Es reicht daher tatsächlich, wenn ein Computer addieren kann. Alles andere 
lässt sich darauf zurückführen.

Statt X*Y zu rechnen, kann man auch X-mal Y aufaddieren:

Beispiel: 3 * 4 = 12
= 4 + 4 + 4 (und zwar genau 3 mal)

5 * 2 = 10
= 2 + 2 + 2 + 2 + 2 (und zwar genau 5 mal)

Schreibe ein Programm, das zwei eingegebene Zahlen multipliziert ohne den Befehl mul zu 
benutzen (also nur mit add). Das Ergebnis soll im Ausgaberegister ausgegeben werden.

Lösung Multiplikation durch Addition

Kommentar: Einführung der Zählschleife mit Hilfe eines Counters. Als Zählschleife gehören 
daher alle gelb markierten Zeilen mit zur Wiederholungsanweisung. Evtl. Hinweis, dass 
eine Zählschleife ebenso nichts anderes ist als eine Endlosschleife mit Zähler und break. 

in $64
in $65
ld 0
st $66 ;counter
ld 0
st $67 ;ergebnis
ld $66 :M1
cmp $64
jeq M2
ld $67
add $65
st $67
ld $66
add 1
st $66
jmp M1
out $67 :M2
end

in $64
in $65
ld 0
st $66 ;counter
ld 0
st $67 ;ergebnis
ld $66 :M1
cmp $64
jeq M2
ld $67
add $65
st $67
ld $66
add 1
st $66
jmp M1
out $67 :M2
end

!

ld 0
st $67 ;ergebnis 
in $64
in $65
ld 0
st $66 ;counter
ld $66 :M1
cmp $64
jeq M2
ld $67
add $65
st $67
ld $66
add 1
st $66
jmp M1
out $67 :M2
end

!



Aufgabe Potenzen

Schreibe ein Programm, das die Potenz von 2 hoch X berechnet und im Ausgaberegister 
anzeigt. X soll die Eingabe sein. 

Beispiel: X = 4 20 Ausgabe 1
21 Ausgabe 2
22 Ausgabe 4
23 Ausgabe 8
24 Ausgabe 16

Hinweis: Die Zahl 2 verdoppelt sich dabei also von Schritt zu Schritt.

Lösung Potenzen

in $64
ld 1
st $65
ld $64 :M2
cmp 0
jeq M1
sub 1
st $64
ld $65
mul 2
st $65
jmp M2
out $65 :M1
end

in $64
ld 1
st $65
ld $64 :M2
cmp 0
jeq M1
sub 1
st $64
ld $65
mul 2
st $65
jmp M2
out $65 :M1
end

!



Aufgabe Fakultät

Die Fakultät der Zahl X berechnet sich als Produkt aller Zahlen von 1 bis X. Die Fakultät 
wird mit dem Ausrufezeichen gekennzeichnet;

Beispiel: 6! (sprich: sechs Fakultät) = 720,   da 1 * 2 * 3 * 4 * 5 * 6 = 720

4! = 24, da 1 * 2 * 3 * 4 = 24

Schreibe ein Programm, das die Fakultät einer ins Eingaberegister eingegebenen Zahl 
berechnet und im Ausgaberegister ausgibt!

Lösung Fakultät

;Alternative

ld 1
st $65
in $64
ld $64
cmp 0 :M2
jeq M1
mul $65
st $65
ld $64
sub 1
st $64
jmp M2
out $65 :M1
end

in $65
ld 1
st $64
st $66
ld $66 :Marke1
mul $64
st $66
ld $64
cmp $65
jeq Marke2
add 1
st $64
jmp Marke1
out $66 :Marke2
end



Aufgabe Suppentemperatur

Eine Suppe, frisch vom Herd gebracht, habe eine Temperatur von 92 Grad Celsius. Die 
Zimmertemperatur betrage 19 Grad Celsius. In jeder Minute verringert sich die 
Suppentemperatur um 10% der Differenz zur Zimmertemperatur. 

Die Suppentemperatur ist daher über die Zeit gesehen wie folgt:

92, 85, 79, 73, 68, 64, 60, 56, 53, 50, 47, 45, 43, 41, 39, ... usw.

Schreibe ein Programm, das die Suppentemperatur in Minutenabständen angibt. Gestalte das 
Programm von Anfang an so, dass zwei Werte eingegeben werden können (Suppentemperatur und 
Raumtemperatur) und dass alle minütlichen Zwischenergebnisse im Ausgaberegister 
ausgegeben werden.

Lösung Suppentemperatur

in $64 ;Temp_Suppe
in $65 ;Temp_Raum
ld $64 :M1
sub $65
div 10
st $66
ld $64
sub $66
st $64
out $64
jmp M1
end



Aufgabe Collatz-Problem

Schreibe ein Programm für ein echtes Problem der Mathematik:

• Beginne mit irgendeiner natürlichen Zahl n > 0, die der Benutzer eingibt.
• Ist n gerade, so nimm als nächstes n/2,
• Ist n ungerade, so nimm als nächstes 3n + 1.
• Wiederhole die Vorgehensweise mit der erhaltenen Zahl.

So erhält man zum Beispiel für die Startzahl n = 19 die Folge

19, 58, 29, 88, 44, 22, 11, 34, 17, 52, 26, 13, 40, 20, 10, 5, 16, 8, 4, 2, 1

Anscheinend mündet jede Folge mit n > 0 früher oder später in den Zyklus 4, 2, 1, 
unabhängig davon, welche Startzahl n man probiert hat. Die Collatz-Vermutung lautet daher 
„Jede so konstruierte Zahlenfolge mündet in den Zyklus 4, 2, 1, egal, mit welcher 
natürlichen Zahl n > 0 man beginnt“. Trotz zahlreicher Anstrengungen gehört diese 
Vermutung noch immer zu den ungelösten Problemen der Mathematik.

Finde die längste Zahlenfolge bei geschickter Startwertwahl!

Lösung Collatz-Problem

ld 0
st $65 ; anzahl
in $64
ld $64 :M1
cmp 1
jeq M2
mod 2
cmp 0
jeq JA
ld $64 ; NEIN
mul 3 ; NEIN
add 1 ; NEIN
jmp M3
ld $64 :JA
div 2 ; JA
st $64 :M3
out $64
ld $65
add 1
st $65
jmp M1
end :M2



Aufgabe Fibonacci-Folge

Die Fibonacci-Folge ist die unendliche Folge von natürlichen Zahlen, die mit 0, 1 beginnt 
und in der die Summe zweier aufeinanderfolgender Zahlen die unmittelbar danach folgende 
Zahl ergibt:

0, 1, 1, 2, 3, 5, 8, 13, 21, 34, 55, 89, ... usw

Die darin enthaltenen Zahlen heißen Fibonacci-Zahlen. Benannt ist die Folge nach Leonardo 
Fibonacci, der damit im Jahr 1202 das Wachstum einer Kaninchenpopulation beschrieb.

Man kann also mit 0 und 1 starten und die darauf folgende Zahl durch Addition berechnen. 
Das geht dann immer so weiter:

0 + 1 = 1
1 + 1 = 2
1 + 2 = 3
2 + 3 = 5
3 + 5 = 8

...usw...

Schreibe ein Programm, das mit den Startwerten 0 und 1 anfängt und im Ausgaberegister 
immer automatisch das nächste Folgenmitglied ausgibt.

Lösung Fibonacci-Folge

ld 0
st $64
ld 1
st $65
;Endlosschleife
ld $64 :M1
add $65
st $66
ld $65
st $64
ld $66
st $65
out $65
jmp M1
end

Kommentar: Es bieten sich hier z.B. Referate über den goldenen Schnitt oder Fibonacci-
artige Vorkommen in der Natur etc. an. 



Aufgabe Zufallszahlengenerator

Zufallszahlen können im Computer nicht zufällig entstehen. Es muss dafür einen 
Algorithmus geben, der Zahlen produziert, die scheinbar zufällig aussehen. Ein solcher 
Algorithmus ist der des linearen Kongruenzgenerators, der wirklich in einigen 
Programmiersprachen zur Erzeugung zufälliger Zahlen zum Einsatz kommt:

Man wählt möglichst geschickte ganzzahlige Werte für die folgenden Variablen (hier ist 
Ausprobieren oder Nachdenken angesagt):

• m (z.B. 9)
• a (z.B. 4)
• c (z.B. 1)

Dann startet man, indem man für die Zahl X zum Zeitpunkt n (also für Xn) einen Startwert 
festlegt. (z.B. 0)

a) Schreibe nun ein Programm, das mit den o.g. Werten Zufallszahlen nacheinander ins 
Ausgaberegister schreibt und ausgibt. 

b) Probiere danach andere Werte für m, a und c aus. Findest Du „bessere“ oder 
vielleicht auch „schlechtere“ Werte, die bessere oder schlechtere „Zufallszahlen“ 
erzeugen? 

c) Diskutiere danach, inwiefern sich Zufallszahlen überhaupt mit dem Computer erzeugen 
lassen.

d) Entwickle eine Idee, wie man echte Zufallszahlen auf den Computer bekommen könnte.
e) Schaue erst danach auf www.random.org vorbei.

EXKURS zum Modulo:

Der Modulo ist der 
ganzzahlige Rest einer 
ganzzahligen Division, 
hier also immer der Rest:

8 / 5 = 1, Rest 3 
8 / 4 = 2, Rest 0
8 / 3 = 2, Rest 2

also:

8 mod 5 = 3
8 mod 4 = 0
8 mod 3 = 2

usw.

Mit der Formel:

�  

kann nun der nächste darauffolgende Wert Xn+1 berechnet 
werden.

Beispiel: m = 9, a = 4, c = 1, X0 = 0

(4*0 + 1) mod 9 = 1
(4*1 + 1) mod 9 = 5
(4*5 + 1) mod 9 = 3
(4*3 + 1) mod 9 = 4
(4*4 + 1) mod 9 = 8
(4*8 + 1) mod 9 = 6
(4*6 + 1) mod 9 = 7
(4*7 + 1) mod 9 = 2
(4*2 + 1) mod 9 = 0

(4*0 + 1) mod 9 = 1 
jetzt geht es wieder von vorne los...



Lösung Zufallszahlengenerator

ld 9 ;m
st $64
ld 4 ;a
st $65
ld 1 ;c
st $66
ld 0
st $67
;start prng
ld $67 :M1
mul $65
add $66
mod $64
st $67
out $67
jmp M1
end

Kommentar: Zahlreiche Anknüpfungspunkte zum Themenkomplex Informatik, Mensch und 
Gesellschaft, da Computer per se keine Zufallszahlen erzeugen können. 



Aufgabe Restfrei durch 1 bis 9

Schreibe ein Programm, um die kleinste Zahl zu ermitteln, die ohne Rest (!) gleichzeitig 
durch 2, 3, 4, 5, 6, 7, 8 und 9 teilbar ist. Die gefundene Zahl soll dann im 
Ausgaberegister ausgegeben werden.

Hinweis: Das fertige Programm kann durchaus eine halbe Stunde laufen bis die Zahl
gefunden wurde.

Lösung Restfrei durch 1 bis 9

Kommentar: Die kleinste Zahl, die restfrei durch 1 bis 9 teilbar ist, ist 2520.

Links eine Lösung mit ineinander geschachtelten Schleifen, rechts eine Lösung mit 
einfachem Testen (es werden nur notwendige Divisoren überprüft, daher kommt man mit dem 
Speicher aus).  

ld 1
st $64 ; zahl

jmp FResCount :Sch1
ld $64 :Sch2
mod $65
cmp 0
jeq FIncCount
jlt FIncNum
jgt FIncNum

;counter auf 1
ld 1 :FResCount
st $65
jmp Sch2

;counter erhöhen
ld $65 :FIncCount
add 1
cmp 10
jeq FOut
st $65
jmp Sch2

;zahl erhöhen
ld $64 :FIncNum
add 1
st $64
jmp Sch1

;ergebnis ausgeben
out $64 :FOut
end

; Alternative

ld 1
st $64
ld $64 :M1

mod 8
cmp 0
jgt M2

ld $64
mod 9
cmp 0
jgt M2

ld $64
mod 6
cmp 0
jgt M2

ld $64
mod 7
cmp 0
jgt M2

ld $64
mod 5
cmp 0
jgt M2
jmp M3

ld $64 :M2
add 1
st $64
jmp M1

ld $64 :M3
out $64
end



Aufgabe Quadriere und Summiere

Wenn man die Ziffern einer ganzen mehrstelligen Zahl einzeln aufteilt, dann einzeln 
quadriert und dann zusammenaddiert, erhält man eine neue Zahl. Wiederholt man das 
mehrfach, bekommt man eine Reihe, die irgendwann bei 1 oder bei 89 in eine Endlosschleife 
mündet. 

Beispiel: 44 (42 + 42 = 32)
32 (32 + 22 = 13)
13 (12 + 32 = 10)
10 (12 + 02 = 1) 
1 (12 = 1) usw. usw.

85, 89, 145, 42, 20, 4, 16, 37, 58, 89

Schreibe ein Programm, das eine beliebige Zahl als Eingabe erwartet und in Adresse $64 
immer automatisch das nächste Folgenglied ausgibt.

Fertige eine Tabelle an für die Zahlen von 1 bis 20 und bestimme, ob diese in eine 89-
Schleife oder in eine 1-Schleife münden.

Tipp: Du wirst fast den ganzen Speicher brauchen. Denke auch daran, dass die 
Zwischenergebnisse bis zu vier Stellen haben können. 

Lösung Quadriere und summiere

in $64
ld $64 :M1

div 1000
st $68
mul $68
st $68

ld $64
mod 1000
div 100
st $69
mul $69
st $69

ld $64
mod 1000
mod 100
div 10
st $70
mul $70
st $70

ld $64
mod 1000
mod 100
mod 10
st $71
mul $71
st $71

add $70
add $69
add $68
st $64
jmp M1

end


